PharmaSUG 2015 - Paper QT15

A Macro to Easily Generate a Calendar Report

Ting Sa, Cincinnati Children’s Hospital Medical Center, Cincinnati, OH

ABSTRACT

This paper introduces a macro that generates a calendar report in two different formats. The first format displays the entire month in one plot, which is called a month-by-month calendar report. The second format displays the entire month in one row and is called an all-in-one calendar report. To use the macro, you just need to prepare a simple data set that has three columns: one column identifies the ID, one column contains the date, and one column specifies the notes for the dates. On the generated calendar reports, you can include notes and add different styles to certain dates. Also, the macro provides the option for you to decide whether those months in your data set that do not contain data should be shown on the reports.

INTRODUCTION

A typical scenario in clinical research involves tracking daily adherence, such as daily adherence to medication intake, or daily adherence to treatment instructions. Viewing these data in a calendar format can be a very useful graphic tool. One common calendar format displays each calendar month in one plot. We refer to this format as a “month-by-month” calendar report. Figure 1 is a sample figure of this “month-by-month” calendar report:

![Figure 1. A Sample Figure for the “month-by-month” Calendar Report](image)

Although there are SAS codes already available to generate calendar reports [1] [2], you still need to update those existing codes for your own needs. The macro presented in this paper simplifies most of the steps. The only must-do work left for you is to prepare a SAS input data set that has three columns: one column identifies the ID; one column includes the dates; and one column specifies the notes. Figure 3 is a sample figure of an input dataset “test” saved in the “work” library:
Figure 3 A Sample Input Data Set “test” Saved in the “work” Library

Once you’ve prepared the data set, you are ready to use the macro in this paper to generate a calendar report. Let’s take a look at this macro.

THE MACRO PARAMETERS

You can find the whole codes of the macro at the end of the paper. Below is the layout of the macro in order to show you the parameters you need to pass to the macro:

```sas
%macro calendar_report(libname=, dataname=, idvar=, datevar=, notevar=,
                      calendarformat=, continuous_month_flag=);
/*the codes are at the end of the paper*/
%mend;
```

- The “libname” is used to indicate the library name for the input dataset.
- The “dataname” is used to indicate the input SAS dataset name.
- The “idvar” is used to indicate the variable name that saves the ID data.
- The “datevar” is used to indicate the variable name that saves the dates data.
- The “notevar” is used to indicate the variable name that saves the date notes data.
- The “calendarformat” is used to indicate the calendar format you want to generate. You can pass 1 to select the “month-by-month” format; pass 2 to select the “all-in-one” format.
- The macro parameter “continuous_month_flag” is used to tell the macro for those months that don’t have data in your data set, whether they should appear in the calendar reports or not. If you pass 0 to the “continuous_month_flag” macro variable, the calendar report will only include the months that have the dates in the input data set. If you pass 1 to this parameter, you will include all the dates that are within the dates range for an ID in the dataset.

For example, in the above sample data set, ID 1 has data for January 2011 and March 2011. If the value of “continuous_month_flag” is 0, the calendar report for ID 1 will only include January 2011 and March 2011 in the calendar report for ID 1. Figure 4 is a screenshot for a “month-by-month” calendar report for ID 1:
A Macro to Easily Generate a Calendar Report, continued

If the value of "continuous_month_flag" is 1, the calendar report will include all the dates that are within the dates range for ID 1 in the data set. In the sample dataset, note that the dates for ID 1 are from January 2011 to March 2011. Therefore, the month of February 2011 will also be included in the calendar report, even if no dates are available for this month in the sample data set.

**Figure 5** is the screenshot for an “all-in-one” calendar report for ID 1.

### CALL THE MACRO:

Use the sample data set in the paper, to generate the “month-by-month” format calendar report, and include all the dates within the date range for an ID in the input date set, your macro will look like the following:

```sas
%calendar_report(libname=work, dataname=test, idvar=id, datevar=date, notevar=note, calendarformat=1, continuous_month_flag=1);
```

To generate the “all-in-one” format calendar report, and display only the months available in the input data set, your macro will look like the following:

```sas
%calendar_report(libname=work, dataname=test, idvar=id, datevar=date, notevar=note, calendarformat=2, continuous_month_flag=0);
```
ADD THE STYLES TO THE CALENDAR

You can add different styles to the calendar reports. You can add styles by adding the styles directly to the proc report codes in the macro [3] or use the SAS template [4] or use the SAS inline formatting functions [5]. Please check the reference to find the details about how to add the styles. In this paper, we use the SAS inline formatting functions to add the foreground and background colors to the reports.

To use the inline formatting functions to add the foreground and background colors, you just need to modify your input dataset as shown in Figure 6 and pass this new input dataset to the macro. The resulting calendar will have blue foreground color and red/green/yellow background colors depending on the values in the note variable.

---

**Figure 6 An Example of Using Inline Formatting Functions to Add the Foreground and Background Colors**

<table>
<thead>
<tr>
<th>id</th>
<th>date</th>
<th>note</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>2011-01-01</td>
<td>“style [foreground=blue background=red] less exercise”</td>
</tr>
<tr>
<td>2</td>
<td>2011-01-01</td>
<td>“style [foreground=blue background=red] less exercise”</td>
</tr>
<tr>
<td>3</td>
<td>2011-01-01</td>
<td>“style [foreground=blue background=red] less exercise”</td>
</tr>
<tr>
<td>4</td>
<td>2011-02-01</td>
<td>“style [foreground=blue background=red] less exercise”</td>
</tr>
<tr>
<td>5</td>
<td>2012-04-20</td>
<td>“style [foreground=blue background=red] less exercise”</td>
</tr>
<tr>
<td>6</td>
<td>2012-04-21</td>
<td>“style [foreground=blue background=green] normal exercise”</td>
</tr>
<tr>
<td>7</td>
<td>2012-04-22</td>
<td>“style [foreground=blue background=yellow] more exercise”</td>
</tr>
<tr>
<td>8</td>
<td>2012-05-20</td>
<td>“style [foreground=blue background=red] less exercise”</td>
</tr>
<tr>
<td>9</td>
<td>2012-05-21</td>
<td>“style [foreground=blue background=green] normal exercise”</td>
</tr>
<tr>
<td>10</td>
<td>2012-05-22</td>
<td>“style [foreground=blue background=yellow] more exercise”</td>
</tr>
<tr>
<td>11</td>
<td>2013-07-01</td>
<td>“style [foreground=blue background=yellow] more exercise”</td>
</tr>
<tr>
<td>12</td>
<td>2013-07-01</td>
<td>“style [foreground=blue background=red] less exercise”</td>
</tr>
<tr>
<td>13</td>
<td>2013-08-15</td>
<td>“style [foreground=blue background=yellow] more exercise”</td>
</tr>
<tr>
<td>14</td>
<td>2013-08-16</td>
<td>“style [foreground=blue background=red] less exercise”</td>
</tr>
<tr>
<td>15</td>
<td>2013-09-15</td>
<td>“style [foreground=blue background=yellow] more exercise”</td>
</tr>
<tr>
<td>16</td>
<td>2013-09-16</td>
<td>“style [foreground=blue background=red] less exercise”</td>
</tr>
</tbody>
</table>

---

THE MACRO CODES

```sas
%macro calendar_report(libname=, dataname=, idvar=, datevar=, notevar=, calendarformat=, continuous_month_flag=);
  data tmp_data;
    set &libname..&dataname. ;
    Year=year(&datevar.);
    Month=month(&datevar.);
  run;

  %if &continuous_month_flag.=1 %then %do;
    proc sql;
    create table tmp_all_dates as
      select &idvar.,
             intnx("month",min(&datevar.),0) as start_date format=mmddyy10.,
             intnx("month",max(&datevar.),1)-1 as end_date format=mmddyy10.
      from tmp_data
    group by &idvar.
    order by &idvar.
  quit;
  %end;
  %else %if &continuous_month_flag.=0 %then %do;
    proc sql;
    create table tmp_all_dates as
      select distinct &idvar.,
             year,month,intnx("month",date,0,"b") as start_date format=mmddyy10.,
```
A Macro to Easily Generate a Calendar Report, continued

```
intnx("month", date, 0, "e") as end_date format=mmddyy10.
from tmp_data
group by &idvar.,year,month,start_date,end_date
order by &idvar.;
quit;
%end;
data tmp_all_dates;
set tmp_all_dates;
do &datevar. = start_date to end_date;
    format &datevar. mmddyy10.;
    output;
end;
run;
proc sort data=tmp_all_dates;by &idvar. &datevar.;run;
data tmp_data;
merge tmp_all_dates tmp_data;
by &idvar. &datevar.;
Year=year(&datevar.);
Month=month(&datevar.);
Week=intck('week',intnx('month',&datevar.,0),&datevar.);
Weekday=put(&datevar.,weekdate9.);
month_start_date = intnx("month", &datevar., 0, "b");
month_end_date = intnx("month", &datevar., 0, "e");
Day=&datevar.-month_start_date+1;
length new_note $98.;
new_note=cats(day,"^n",&notevar.);
format month_start_date month_end_date date9.;
run;
ods listing close;
ODS ESCAPECHAR= '^';
%if &calendarformat.=1 %then %do;
proc sort data=tmp_data;by &idvar. Year Month Week Weekday;run;
proc transpose data=tmp_data out=tmp_data1;
by &idvar. year month;
id weekday;
var new_note;
run;
options missing=' '; 
ods html body="calendar.html" style=sasweb;
proc report data=tmp_data1 center nowd
style (column)={just=center cellheight=50pt cellwidth=10% just=left vjust=top};
by &idvar. year month;
columns Sunday Monday Tuesday Wednesday Thursday Friday Saturday;
run;
ods html close;
%end;
%if &calendarformat.=2 %then %do;
proc format;
```
value weekdaylabel
1="Sun"
2="Mon"
3="Tue"
4="Wed"
5="Thu"
6="Fri"
0,7="Sat"
;
run;

data tmp_data2;
  set tmp_data;
  array arr_days{1:37} $2. ;
  array arr_notes{1:37} $100.;
  array arr_daynotes{1:37} $256. ;
  retain arr_days arr_notes arr_daynotes;
  month_start_date_weekday=weekday(month_start_date);
  i=day+month_start_date_weekday-1;
  arr_days{i} = day;
  arr_notes{i} = note;
  if not missing(note) then arr_daynotes{i} = new_note;
  else arr_daynotes{i}=arr_days{i};
  if date = month_end_date then do 
    output ;
    call missing(of arr_days{*}, of arr_notes{*}, of arr_daynotes{*}) ;
  end;
  label
  %do i=1 %to 37 ;
    %let weekdayno = %SYSFUNC(mod(&I,7)) ;
    %LET weekdayname =
    %SYSFUNC(putn(&weekdayno,weekdaylabel.));
    arr_daynotes&i = "&weekdayname" 
  %end;
  run;
ods html body="calendar.html" style=sasweb;
proc report data = tmp_data2 nowd ;
  by id;
  column date
    %do i=1 %to 37 ;
      arr_daynotes&i
    %end;
    ;
  define date / display order order=data format=monyy7. ;
    %do i=1 %to 37 ;
      define arr_daynotes&i / display center;
    %end;
  run;
ods html close;
%end;
ods listing;
proc datasets;delete tmp_;run;quit;
%mend;
CONCLUSION
The macro presented in this paper provides an easy way to generate calendar reports. Currently, the macro prints out the calendar reports in html. You can modify the macro to print the results in other output formats, such as rtf and excel. In addition to adding styles to the calendar reports, you can also put date notes in a “pop out” tooltip. In this case, when you hover the mouse over a certain date on the report, a tooltip will pop up to show you the notes for that date. Please check reference [2] for details on adding the tooltip function.

REFERENCES

2. INTOCODE.ORG, 30APR2013. “FTSE 100 Heatmap Calendar with Hover Functionality in SAS”. Available at http://intocode.org/sas/hoverftseheatmap.php


ACKNOWLEDGMENTS
The authors wish to thank the Division of Biostatistics and Epidemiology at Cincinnati Children’s Hospital Medical Center for its support, and particularly Robert Tamer and the SAS Programming Group for their helpful feedback.

CONTACT INFORMATION
Your comments and questions are valued and encouraged. Contact the author at:

Name: Ting Sa
Enterprise: Division of Biostatistics and Epidemiology, Cincinnati Children’s Hospital Medical Center
Address: 3333 Burnet Ave
City, State ZIP: Cincinnati, OH 45229
Work Phone: 513-636-3674
E-mail: ting.sa@cchmc.org

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.